興奮系の数理

電気生理で出てくる膜電位の脱分極ー過分極のダイナミクスは、自然言語による説明だけではわかりにくい。一番単純なFitzHugh-Nagumo型の系のダイナミクスを簡単に再現してみる。uが膜電位、vが不活性化を表す抽象的な変数である。

In[126]:=
```
a = 0.7; b = 0.8; c = 10; dt = 0.01; Ini = 0;
```

In[127]:=
```
f[{u_, v_}] := c (-u^3 / 3 + u - v + Ini);
g[{u_, v_}] := u - b v + a;
```

これは、

関数fに関しては、
* 膜電位は、増加するとさらに増加しやすい、減少するとさらに減少しやすい、というpositive feedbackの性質がある（+uの項）
* 膜電位は有界で、ある値以上や以下にはならない（-u^3/3の項）
* 別の因子vの影響で不活性化される(-vの項）

関数gに関しては
* 膜電位が変化すると、膜電位を不活性化させる因子vが増加する(uの項)

という意味がある。

まずこの系で、数値計算で系のダイナミクスを再現してみる。弱い刺激によって少しだけ平衡点からずれた状態 (u, v)=(-0.7,-0.62)と、強い刺激によって大きく大きくずれた状態(u, v)=(-0.7,-0.62)の２つを考える。

ある短い時間で (u,v) がどれだけ変化するかを定義する関数を作る。

In[129]:=
```
oneStep[{u_, v_}] := {u, v} + dt {f[{u, v}], g[{u, v}]};
```

この関数をこの２つの初期値に繰り返し適用して、数値的に計算してみる。

In[130]:=
```
result1 = NestList[oneStep, {-0.8, -0.62}, 500];
```

In[131]:=
```
result2 = NestList[oneStep, {-0.2, -0.62}, 500];
```

uの時間変化をプロットしてみる。弱い刺激は青、強い刺激は赤で表示してやる。

In[132]:=
```
up1 = ListPlot[Transpose[result1][[1]], PlotRange → All]
```

Out[132]=



In[133]:=
```
up2 = ListPlot[Transpose[result2][[1]], PlotRange → All, PlotStyle → Red]
```

Out[133]=



In[134]:=
```
Show[up1, up2]
```

Out[134]=



強い刺激では脱分極ー過分極反応が起こることがわかると思う。さて、これはなぜこのような差が起こるのだろうか？uだけではなく、(u,v)の軌道を相平面に表示してみる。

In[135]:=
```
p1 = ListPlot[result1, PlotRange → {{-2, 2}, {-2, 2}}]
```

Out[135]=



In[136]:=
```
p2 = ListPlot[result2, PlotRange → {{-2, 2}, {-2, 2}}, PlotStyle → Red]
```

Out[136]=



In[137]:=
```
Show[p1, p2]
```

Out[137]=



刺激が強い方が大回りして元の点に戻って行くことがわかると思う。このダイナミクスを見るために、(u,v)がどの値だったら（u’, v’ ）がどうなっているか（=u,v相平面上でどちらに動くか）をベクトル場で描いてみる。(u,v)がある値を取っているとき、次にどのような値になるかは、その点のベクトルの方向に動いた値を取る。

In[138]:=
```
phasePlotVector =
 VectorPlot[{f[{u, v}], g[{u, v}]}, {u, -2, 2}, {v, -1, 1}, FrameLabel → {"u", "v"}]
```

Out[138]=



このベクトル場をもっと見やすくするために、u’=0, v’=0の境界線（ヌルクラインと呼ぶ）を一緒に描いてみる。この線のどちら側に(u,v)が存在するかで、右に行くか左に行くか（u’=0）もしくは上に行くか下に行くか（v’=0）が分かれる。

In[139]:=
```
phasePlotUNullcline = Plot[-u^3 / 3 + u, {u, -2, 2}]
```

Out[139]=



In[140]:=
```
phasePlotVNullcline = Plot[(u + a) / b, {u, -2, 2}]
```

Out[140]=



すべてを一緒に表示してみる。

In[141]:=
```
phasePlotVectorNullclineXY =
 Show[phasePlotVector, phasePlotUNullcline, phasePlotVNullcline]
```

Out[141]=



これと先ほどの軌道を重ねて表示してみる。

In[142]:=
```
Show[phasePlotVectorNullclineXY, p1, p2]
```

Out[142]=



この系は、u’ = 0 とv’ = 0 が交わる点が一つしか無い。また、この点での固有値はどちらも負になるので、安定平衡点である。u,vをここからずらしてやっても、基本的にはもとの点に戻って来る、という動きしかしない。しかし、uのヌルクラインを超えるまで強い刺激を与えてやると、上の赤で示すような、遠くを巡回して戻ってくるような軌道を取る場合がある。これが、ある程度以上刺激が強いときに全然別の挙動をする理由である。